

BULCOW

BULLS AND COWS GUESSING GAME

Description

In this game, a somewhat advanced version of BAGLES, the idea is that each player (you and the computer) tries to guess a 5-digit number thought up by the opposing player. A BULL is scored for each correct digit in the correct position and a COW for each correct digit but out of position. For example:




Mystery number	51340
Your guess	21734 scores 1 BULL and 2 COWS

In the first sample run, the human player's mystery number was 12345. In the second run, the number was 13579. Notice that on the fourth computer guess, 35719, the human player told the computer 0,5 (i.e., no BULLS and 5 COWS). This was incorrect; it should have been 1,4, but the computer could not know that until two turns later when it realized that the human had given it impossible scores.

By about the fourth or fifth guess, the computer starts taking quite a bit of time to narrow its choices between guesses. Be patient.

Program Author

Geoff Wyvill
Bradford University
Bradford, Yorkshire, England



PROGRAM LISTING

```

5 GOSUB500
10 DIMD(10,4),B(10),C(10),G(10)
15 RANDOMIZE:PRINT:PRINT
20 LETA=0:GOTO200
30 PRINT:PRINT:PRINT:LETJ=0
35 PRINT"YOUR GUESS":INPUTN:LETN=(N+1)/100000
40 FORI=0TO4:LETG(I)=INT(10*N):LETN=10*N-INT(10*N)
41 FORK=0TO1-1:IFG(I)=G(K)GOTO170
42 NEXTK
43 NEXTI
45 LETP=4:LETA=0:GOSUB300
50 PRINTV"BULL":IFV<>1THENPRINT"S"
55 IFV=5THENPRINT" - YOU WIN":GOTO200
60 PRINTW-V"COW":IFW<>V+1THENPRINT"S"
65 IFJ=0THENLETA=1:GOTO200
68 GOSUB400
70 PRINT" - MY GUESS IS "
75 FORI=0TO4:PRINTCHR$(D(I)+48):NEXTI
80 PRINT" MY SCORE":INPUTB(J),C(J):LETC(J)=C(J)+B(J)
81 IFB(J)>1THENIFB(J)<6THENIFC(J)<6THENIFC(J)-B(J)>-1GOTO83
82 PRINT" - RIDICULOUS!!":GOTO 70
83 IFB(J)=4THENIFC(J)=5GOTO82
85 IFB(J)=5THENPRINT" - I WIN - MY NUMBER WAS":GOTO100
90 GOTO35
100 FORI=0TO4:PRINTCHR$(D(I)+48):NEXTI
110 GOTO200
150 PRINT:PRINT"YOU HAVE GIVEN ME IMPOSSIBLE SCORES - GAME SPOILED"
:GOTO 5
170 PRINT"REPEATED DIGITS NOT ALLOWED":GOTO35
200 FORP=0TO4
210 LETD(A,P)=INT(10*RND(1))
220 FORI=0TOP-1:IFD(A,I)=D(A,P)GOTO210
230 NEXTI
240 NEXTP
250 IFA=0GOTO30
260 LETJ=1:GOTO70
300 LETV=0:LETW=0
310 FORI=0TOP:IFD(A,I)=G(I)THENLETV=V+1
320 FORK=0TO4:IFD(A,K)=G(I)THENLETW=W+1
330 NEXTK
340 NEXTI
350 RETURN
400 LETP=0
405 LETG(P)=D(J,P)
410 FORI=0TOP-1:IFG(I)=G(P)GOTO430
415 NEXTI
420 FORA=1TOJ:GOSUB300
425 IFV<B(A)THENIFW<C(A)THENIF4-P>=C(A)-WTHENIF4-P>=B(A)-VGOTO448
430 LETG(P)=G(P)+3:IFG(P)>9THENLETG(P)=G(P)-10
432 IFP=0THENIFG(P)=D(1,0)GOTO150
435 IFG(P)<>D(J,P)GOTO410
440 LETP=P-1:IFP<0THENGOTO150
445 GOTO430
448 NEXTA
450 LETP=P+1:IFP<5GOTO405
455 LETJ=J+1
460 FORI=0TO4:LETD(J,I)=G(I):NEXTI
465 RETURN
500 PRINT:PRINT" BRADFORD UNIVERSITY BULLS AND COWS GAME
"
510 GOTO10
999 END

```

SAMPLE RUN

BRADFORD UNIVERSITY BULLS AND COWS GAME

```

YOUR GUESS? 12345
0 BULLS 2 COWS - MY GUESS IS 68321 MY SCORE? 1.2
YOUR GUESS? 34567
1 BULL 2 COWS - MY GUESS IS 61953 MY SCORE? 0.3
YOUR GUESS? 67890
0 BULLS 3 COWS - MY GUESS IS 94231 MY SCORE? 0.4
YOUR GUESS? 67813
0 BULLS 3 COWS - MY GUESS IS 27319 MY SCORE? 1.2
YOUR GUESS? 24678
3 BULLS 0 COWS - MY GUESS IS 16429 MY SCORE? 1.2
YOUR GUESS? 25679
1 BULL 1 COW - MY GUESS IS 19384 MY SCORE? 2.1
YOUR GUESS? 64378
2 BULLS 2 COWS - MY GUESS IS 15342 MY SCORE? 2.3
YOUR GUESS? 94638
5 BULLS - YOU WIN

```

Our mystery number was 12345
Computer's score is 1 BULL for the 3
and 2 COWS for the 1 and 2.

```

YOUR GUESS? 12345
1 BULL 1 COW - MY GUESS IS 46098 MY SCORE? 0.1
YOUR GUESS? 13579
0 BULLS 2 COWS - MY GUESS IS 79321 MY SCORE? 0.4
YOUR GUESS? 24680
0 BULLS 3 COWS - MY GUESS IS 02173 MY SCORE? 0.3
YOUR GUESS? 35680
0 BULLS 3 COWS - MY GUESS IS 35719 MY SCORE? 0.5
YOUR GUESS? 80345
0 BULLS 2 COWS - MY GUESS IS 91537 MY SCORE? 0.5
YOUR GUESS? 23568
0 BULLS 4 COWS
YOU HAVE GIVEN ME IMPOSSIBLE SCORES - GAME SPOILED

```

Our mystery number was 13579

This clue was incorrect because the 9 is in
the right place. Score should have been 1.4.